
becker-woessmann-aer-2013-analysis.do
use "becker-woessmann-aer-2013-data.dta", clear

xtset kk_id year


* Table 1 - Trends in Church Attendance and in Income, 1886-1911

xtreg f_comm year, fe cluster(kk_id)
xi: xtreg f_comm i.year, fe cluster(kk_id)
xtreg lteainc year, fe cluster(kk_id)
xi: xtreg lteainc i.year, fe cluster(kk_id)


* Table 2 - Income and Church Attendance, 1886-1911

xi: reg lteainc f_comm i.year, cluster(kk_id)
xtreg lteainc f_comm, fe cluster(kk_id)
xi: xtreg lteainc f_comm i.year, fe cluster(kk_id)
xi: xtreg lteainc f_comm L5.f_comm i.year, fe cluster(kk_id)
xi: xtreg f_comm lteainc L5.lteainc i.year, fe cluster(kk_id)

preserve
keep kk_id year teainc lteainc f_comm f_prot1885 sample_bal daylab_urb
reshape wide teainc lteainc f_comm daylab_urb, i(kk_id) j(year)
gen d_f_comm_86_11 =  f_comm1911 - f_comm1886
gen d_lteainc_86_11 =  log(teainc1911) - log(teainc1886)
reg d_lteainc_86_11 d_f_comm_86_11, cluster(kk_id)
* for Table A4: 
reg d_lteainc_86_11 d_f_comm_86_11 if (f_prot1885>.9 & f_prot1885!=.), cluster(kk_id)
* for Table A5: 
reg d_lteainc_86_11 d_f_comm_86_11 if sample_bal, cluster(kk_id)
* for Table A6: 
gen d_f_comm_91_01 =  f_comm1901 - f_comm1891
gen d_ldaylab_92_01 =  log(daylab_urb1901) - log(daylab_urb1891)
reg d_ldaylab_92_01 d_f_comm_91_01, cluster(kk_id)


* Figure A2 - Change in Income and in Church Attendance, 1886-1911

set scheme s1mono
scatter d_lteainc_86_11 d_f_comm_86_11, title("") subtitle("") xtitle("Change in church attendance, 1886-1911") ytitle("Change in ln(teacher income), 1886-1911") xscale(r(-.2 .1)) xlabel(-.2(.1).1)


* Table A1 - Descriptive Statistics

sum f_comm1886 f_comm1891 f_comm1896 f_comm1901 f_comm1906 f_comm1911 d_f_comm_86_11 lteainc1886 lteainc1891 lteainc1896 lteainc1901 lteainc1906 lteainc1911 
sum d_lteainc_86_11 if d_f_comm_86_11!=.
restore

sum f_comm lteainc
sum f_young f_over60 f_fem f_urban f_rw f_ortsgeb f_prot if f_comm!=.


* Table A2 - Income and Church Attendance: Year-by-Year Cross-Sections

foreach year in 1886 1891 1896 1901 1906 1911 {
	reg lteainc f_comm if year==`year', cluster(kk_id)
}


* Table A3 - Income and Church Attendance, 1886

reg lteainc f_comm if f_young!=., cluster(kk_id)
xi: reg lteainc f_comm f_young f_over60 f_fem f_urban f_rw f_ortsgeb f_prot i.provinzkey, cluster(kk_id)
reg f_comm lteainc if f_young!=., cluster(kk_id)
xi: reg f_comm lteainc f_young f_over60 f_fem f_urban f_rw f_ortsgeb f_prot i.provinzkey, cluster(kk_id)


* Table A4 - Restriction to Counties with at least 90 Percent Protestants

xi: reg lteainc f_comm i.year if (f_prot1885>.9 & f_prot1885!=.), cluster(kk_id)
xtreg lteainc f_comm if (f_prot1885>.9 & f_prot1885!=.), fe cluster(kk_id)
xi: xtreg lteainc f_comm i.year if (f_prot1885>.9 & f_prot1885!=.), fe cluster(kk_id)
xi: xtreg lteainc f_comm L5.f_comm i.year if (f_prot1885>.9 & f_prot1885!=.), fe cluster(kk_id)
xi: xtreg f_comm lteainc L5.lteainc i.year if (f_prot1885>.9 & f_prot1885!=.), fe cluster(kk_id)


* Table A5 - Balanced Panel of Counties with Observed Data in All Waves

xi: reg lteainc f_comm i.year if sample_bal, cluster(kk_id)
xtreg lteainc f_comm if sample_bal, fe cluster(kk_id)
xi: xtreg lteainc f_comm i.year if sample_bal, fe cluster(kk_id)
xi: xtreg lteainc f_comm L5.f_comm i.year if sample_bal, fe cluster(kk_id)
xi: xtreg f_comm lteainc L5.lteainc i.year if sample_bal, fe cluster(kk_id)


* Table A6 - Alternative Income Measure: Wages of Day Laborers in 1892 and 1901

xi: reg ldaylab_urb f_comm i.year, cluster(kk_id)
xtreg ldaylab_urb f_comm, fe cluster(kk_id)
xi: xtreg ldaylab_urb f_comm i.year, fe cluster(kk_id)
xi: xtreg ldaylab_urb f_comm L5.f_comm i.year, fe cluster(kk_id)


* Figure A1 - Income and Church Attendance, 1911

scatter lteainc f_comm if year==1911, title("") subtitle("") xtitle("Church attendance") ytitle("ln(teacher income)") xscale(r(0 1)) xlabel(0(.5)1)

If you change your Stata directory to the path where you have saved  


the dataset (becker-woessmann-aer-2013-data.dta), then running  


the do-file (becker-woessmann-aer-2013-analysis.do) will produce  


all results of the paper and the appendix. Comparison with the paper  


text makes the dataset self-explanatory. 


 






